SOLUTION OF SCHRODINGER EQUATION FOR GROUND STATE OF HYDROGEN ATOM FOR SCREEENED COLOMB POTENTIAL

The shielding effect describes the attraction between an electron and nucleus in any atom with more then one electron shell. Shielding effect can be define as a reduction in the effective nuclear charge on the electron cloud, due to a difference in the attraction force of the electrons on the nucleus it is also referred to as the screening effect (or) atomic shielding.

In hydrogen, or any other atom in group 1A of the periodic table (those with only one valence electron), the force on the electron is just as large as the electromagnetic attraction from the nucleus. However, when more electrons are involved, each electron (in the n-shell) experiences not only the electromagnetic attraction from the positive nucleus, but also repulsion forces from other electrons in shells from 1 to n. This cause the net force on electrons in outer shell to be significantly smaller in magnitude; therefore, these electrons are not as strongly bonded to the nucleus as electrons closer to the nucleus. The shielding theory also contributes to the explanation of why valency-shell electron are more easily removed from the atom.

The potential energy for screening Coulomb potential can be represent as,

$$U_e = \frac{-Ze^2}{r} e^{\frac{-r}{a}}$$

Where,
a = Constant = 3, 5, 7 Armstrong
For this potential radial Schrodinger's equation is written by,

$$-\frac{\hbar^2}{2m_e}\frac{d^2R}{dr^2} + \left(\frac{-Ze^2}{r} e^{\frac{-r}{a}} + \frac{\hbar^2}{2m_e}\frac{\ell(\ell+1)}{r^2}\right)R = ER$$

Now, using the matrix method Schrodinger's equation can be solved. First, the radium matrix can be formed by,

$$r = [r_1, r_2, r_3, \ldots\ldots\ldots, r_n]_{1\times n}$$

$$r_1 = r_{max}$$

$$r_n = r_{min}$$

The potential energy matrix is given by,

$$U(r) = [U_1, U_2, U_3, \dots\dots\dots\dots, U_n]$$

Where, the Potential energy in term of radius can be define as,

$$U_n = \frac{-Ze^2}{r} e^{\frac{-r}{a}} + \frac{l(l+1)}{2m \cdot r_n^2}$$

The Schrodinger Equation in matrix form is given by,

$$[H] = [K] + [U]$$

$$[H] = \frac{-\hbar^2}{2m \cdot (dr)^2} \begin{bmatrix} -2 & 1 & 0 \\ 1 & -2 \cdots & 0 \\ 0 & 1 & 0 \\ \vdots & \ddots & \vdots \\ 0 & 0 \cdots & -2 \end{bmatrix}_{(n-2)\times(n-2)} + \begin{bmatrix} U_1 & 0 & 0 \\ 0 & U_2 \cdots & 0 \\ 0 & 0 & 0 \\ \vdots & \ddots & \vdots \\ 0 & 0 \cdots & U_{n-2} \end{bmatrix}_{(n-2)\times(n-2)}.$$

Thus eigenvalues and eigenfunctions of [H] matrix will be give us the energy eigenvalues and energy eigenfunctions. These energy eigenfunction and probability density can be plotted against the radius.

SCILAB CODE TO SOLVE SCHRODINGERS EQUATION FOR GROUND STATE OF HYDROGEN ATOM FOR SCREENED COULNB POTENTIAL

```
clear
clc

//INPUTS

a=input('Enter the value of a:(3,5 or 7 Armstrong):')
r_max=10
L=input('enter the value of l:')
Z=1
pqn=input('enter the value of n(n>l):')
num=1301

//CONSTANTS
//c=speed of light

hbar=1973
e=3.795
me=0.511e6

//POTENTIAL WELL
//potential energy in electon volts(eV)

r_min=1e-5
r=linspace(r_min,r_max,num)
dr=r(2)-r(1)
dr2=dr^2

//Screened Colomb term
K=-Z*e*e*exp(-r/a)
U_c=K./r

//Angular momentum term
U_L=(hbar^2*L*(L+1)/(2*me))./r.^2

//Effective potential energy
U=U_c+U_L
for cn=1:(num-2)
   U_matrix(cn,cn)=U(cn+1)
end

//SOLVE SCHRODINGER EQUATION
//MAke 2ndary Derivative Matrix
off=ones(num-3,1)
SD_matrix=(-2*eye(num-2,num-2)+diag(off,1)+diag(off,-1))/dr2

//Make KE Matrix
K_matrix=-hbar^2/(2*me)*SD_matrix

//Make Hamiltonian matrix
H_matrix=K_matrix+U_matrix

//Find Eigenvalues E_n and Eigenfunction psi_N
[e_funct,e_values]=spec(H_matrix)

//All Eigenvalues 1,2,...n where E_N<0
//DISPLAYING ReSULTS(Eigenvalues)
En=diag(e_values)
En=(En(En<0))'
```

```
disp('Quantum state, Energy Eigenvalues(in eV):')
for i=1:length(En)
printf('%i %f\n',i,En(i))
end
nl=pqn-L
printf('Energy Eigenvalue(in eV) for Quntum stae no.%i:(l=%i)\n%f',nl,L,En(nl))

//Normalisation of eigenfunctions
for n=1:nl
   psi(:,n)=[0;e_funct(:,n);0]
   area=inttrap(r,(psi(:,n).*psi(:,n))')
   psi(:,n)=psi(:,n)/sqrt(area)
   prob(:,n)=psi(:,n).*psi(:,n)
end

//PLOTTING RESULTS(Eigenfunctions and probability)

figure(1)
xtitle('probability density vs radius by C-RAY')
xlabel('Radius in Armstrong')
ylabel('Probabilty Density per meter')
xgrid
plot(r',prob(:,nl),'b')

figure(2)
xtitle('Wave function vs radius by C-RAY')
xlabel('Radius in Armstrong')
ylabel('Wavefunction')
xgrid
plot(r',psi(:,nl),'b')
```

RESULTS AND DISCUSSIONS

After executing the given Scilab code we get the energy eigenvalues and corresponding eigenfunctions the ground state.

FOR GROUND STATE:

Enter for a= 3 Armstrongs
Enter the value of l: 0
Enter the value of n(n>l): 1

Quantum state	Energy Eigenvalues(in eV)
1	-9.385503
2	-0.482507

Energy Eigenvalue (in eV) for Quantum state no. 1:(l=0)
-9.385503

GRAPH-1



GRAPH-2



CONCLUSION:

The energy of the ground state is -9.385503 eV respectively.