

SOLUTION OF SCHRODINGER EQUATION FOR VIBRATION OF HYDROGEN MOLECULE

The Morse potential is a convenient interatomic model for the potential energy of a diatomic molecule. It is a better approximation for the vibrational structure of the molecule than QHO (Quantum Harmonic Oscillator) because it explicitly includes the effects of bond breaking, such as the existence of unbound states.

The potential energy due to Morse potential is given as,

$$U(r) = D(e^{-2\alpha r'} - 2e^{-\alpha r'})$$

Where,

$$r' = \frac{r - r_0}{r}$$

$D = 0.755501 \text{ eV}$ = Depth of potential well

$\alpha = 1.44$ = Constant = Controls the 'Width' of the potential (the smaller it is, the larger the well)

$r_0 = 0.131349 \text{ \AA}$ = Equilibrium bond distance

r = Distance between the atoms

Here, mass of the system is taken as $m = 940 \times 10^6 \text{ eV}/c^2$

Now, using the matrix method Schrodinger's equation can be solved. First, the radius matrix can be formed by,

$$r = [r_1, r_2, r_3, \dots, r_n]_{1 \times n}$$

$$r_1 = r_{\max}$$

$$r_n = r_{\min}$$

The potential energy matrix is given by,

$$U(r) = [U_1, U_2, U_3, \dots, U_n]$$

Where, the Potential energy in term of radius can be define as,

$$U_n = D(e^{-2ar'} - 2e^{-ar'}) + \frac{l(l+1)}{2m \cdot r_n^2}$$

The Schrodinger Equation in matrix form is given by,

$$[H] = [K] + [U]$$

$$[H] = \frac{-\hbar^2}{2m \cdot (dr)^2} \begin{bmatrix} -2 & 1 & 0 \\ 1 & -2 \cdots & 0 \\ 0 & 1 & 0 \\ \vdots & \ddots & \vdots \\ 0 & 0 \cdots & -2 \end{bmatrix}_{(n-2) \times (n-2)} + \begin{bmatrix} U_1 & 0 & 0 \\ 0 & U_2 \cdots & 0 \\ 0 & 0 & 0 \\ \vdots & \ddots & \vdots \\ 0 & 0 \cdots & U_{n-2} \end{bmatrix}_{(n-2) \times (n-2)}$$

Thus eigenvalues and eigenfunctions of [H] matrix will be give us the energy eigenvalues and energy eigenfunctions. These energy eigenfunction and probability density can be plotted against the radius.

SCILAB CODE TO SOLVE SCHRODINGERS EQUATION FOR GROUND STATE AND FIRST EXCITED STATE OF HYDROGEN ATOM

```

clear
clc

//INPUTS

r_max=7 //maximum radius(Range) Default=1 Armstrong(A)
L=input('enter the value of l:') //Orbital angular momentum quantum no
Z=1 //Atomic no. or total charge
pqn=input('enter the value of n(n>l):') //Principal quantum number,
//for ground state pqn=1
//1st Excited state pqn
num=1501 //number of odd data points Default 1501

//CONSTANTS
//c=Speed of light

hbar=1973
me=940e6
D=0.755501
r0=0.131349
a=1.44

//POTENTIAL WELL
//potential energy in electron volts(eV)

r_min=1e-5 //Default 1e-5 in Armstrong(A)
r=linspace(r_min,r_max,num)
dr=r(2)-r(1)
dr2=dr^2

r1=(r-(r0*ones(1,num)))./r //Making r1 matrix

//Morse Potential
K=D*(exp(-2*a.*r1)-2*exp(-a.*r1))
U_c=K

//Angular momentum term
U_L=(hbar^2*L*(L+1)/(2*me))./r.^2

//Effective potential energy
U=U_c+U_L
for cn=1:(num-2)
    U_matrix(cn,cn)=U(cn+1)
end

//SOLVE SCHRODINGER EQUATION
//MAke 2ndary Derivative Matrix
off=ones(num-3,1)
SD_matrix=(-2*eye(num-2,num-2)+diag(off,1)+diag(off,-1))/dr2

//Make KE Matrix
K_matrix=-hbar^2/(2*me)*SD_matrix

//Make Hamiltonian matrix
H_matrix=K_matrix+U_matrix

//Find Eigenvalues E_n and Eigenfunction psi_N
[e_funct,e_values]=spec(H_matrix)

```

```

//All Eigenvalues 1,2,...n where E_N<0
//DISPLAYING ReSULTS(Eigenvalues)
En=diag(e_values)
En=(En(En<0))'

disp('Quantum state, Energy Eigenvalues(in eV):')
for i=1:length(En)
    printf('%i %f\n',i,En(i))
end
nl=pqn-L
printf('Energy Eigenvalue(in eV) for Quntum stae no.%i:(l=%i)\n%f',nl,L,En(nl))

//Normalisation of eigenfunctions
for n=1:nl
    psi(:,n)=[0;e_funct(:,n);0]
    area=inttrap(r,(psi(:,n).*psi(:,n)))
    psi(:,n)=psi(:,n)/sqrt(area)
    prob(:,n)=psi(:,n).*psi(:,n)
end

//PLOTTING RESULTS(Eigenfunctions and probability)

figure(1)
xtitle('probability density vs radius by C-RAY')
xlabel('Radius in Armstrong')
ylabel('Probabilty Density per meter')
xgrid
plot(r',prob(:,nl),'b')

figure(2)
xtitle('Wave function vs radius by C-RAY')
xlabel('Radius in Armstrong')
ylabel('Wavefunction')
xgrid
plot(r',psi(:,nl),'b')

```

RESULTS AND DISCUSSIONS

After executing the given Scilab code we get the energy eigenvalues and corresponding eigenfunctions the ground or the lowest vibrational energy of hydrogen molecule.



FOR GROUND STATE:

Enter the value of l : 0

Enter the value of $n(n>l)$: 1

Quantum state	Energy Eigenvalues(in eV)
1	-0.482423
2	-0.373236
3	-0.344189
4	-0.332592
5	-0.326834
6	-0.323246
7	-0.319397
8	-0.314388
9	-0.308273
10	-0.301127
11	-0.292996
12	-0.283910
13	-0.273889
14	-0.262948
15	-0.251098
16	-0.238349
17	-0.224707
18	-0.210178
19	-0.194768
20	-0.178479
21	-0.161316
22	-0.143281
23	-0.124377
24	-0.104607
25	-0.083971
26	-0.062472
27	-0.040112
28	-0.016892

Energy Eigenvalue (in eV) for Quantum state no. 1: ($l=0$)
-0.482423

GRAPH -1**GRAPH -2****CONCLUSION:**

The lowest vibrational energy of hydrogen molecule is found to be -0.482423 eV.