SOLUTION OF SCHRODINGER EQUATION FOR GROUND STATE AND FIRST EXCITED STATE OF HYDROGEN ATOM

The hydrogen atom consist of a single electron orbiting around a nucleus, containing one single proton, under the influence of the Cpulombic field. From the viewpoint of quantum mechanics, electron is represent by the potential well of coulomb field. The potential energy Ue is given by,

$$U_e = \frac{-Ze^2}{r}$$

For this potential radial Schrodinger's equation is written by,

$$-\frac{\hbar^2}{2m_e}\frac{d^2R}{dr^2}+\left(\frac{-Ze^2}{r}+\frac{\hbar^2}{2m_e}\frac{\ell(\ell+1)}{r^2}\right)R = ER$$

Now, using the matrix method Schrodinger's equation can be solved. First, the radium matrix can be formed by,

$$r = [r_1, r_2, r_3, \dots\dots\dots, r_n]_{1\times n}$$

$$r_1 = r_{max}$$

$$r_n = r_{min}$$

The potential energy matrix is given by,

$$U(r) = [U_1, U_2, U_3, \dots\dots\dots\dots, U_n]$$

Where, the Potential energy in term of radius can be define as,

$$U_n = \frac{-Ze^2}{r_n} + \frac{l(l+1)}{2m\,.\,r_n^2}$$

The Schrodinger Equation in matrix form is given by,

$$[H] = [K] + [U]$$

$$[H] = \frac{-\hbar^2}{2m.(dr)^2}\begin{bmatrix} -2 & 1 & 0 \\ 1 & -2 \cdots & 0 \\ 0 & 1 & 0 \\ \vdots & \ddots & \vdots \\ 0 & 0 \cdots & -2 \end{bmatrix}_{(n-2)\times(n-2)} + \begin{bmatrix} U_1 & 0 & 0 \\ 0 & U_2 \cdots & 0 \\ 0 & 0 & 0 \\ \vdots & \ddots & \vdots \\ 0 & 0 \cdots & U_{n-2} \end{bmatrix}_{(n-2)\times(n-2)}.$$

Thus eigenvalues and eigenfunctions of [H] matrix will be give us the energy eigenvalues and energy eigenfunctions. These energy eigenfunction and probability density can be plotted against the radius.

SCILAB CODE TO SOLVE SCHRODINGERS EQUATION FOR GROUND STATE AND FIRST EXCITED STATE OF HYDROGEN ATOM

```
clear
clc

//INPUTS

r_max=10
L=input('enter the value of l:')
Z=1
pqn=input('enter the value of n(n>l):')
num=1301

//CONSTANTS
//c=speed of light

hbar=1973
e=3.795
me=0.511e6

//POTENTIAL WELL
//potential energy in electon volts(eV)

r_min=1e-5
r=linspace(r_min,r_max,num)
dr=r(2)-r(1)
dr2=dr^2

//couloumb term
K=-Z*e*e
U_c=K./r

//angular momentum term
U_L=(hbar^2*L*(L+1)/(2*me))./r.^2

//Effective potential energy
U=U_c+U_L
for cn=1:(num-2)
   U_matrix(cn,cn)=U(cn+1)
end

//solve schrodiger equations
//Make second derivative matrix

off=ones(num-3,1)
SD_matrix=(-2*eye(num-2,num-2)+diag(off,1)+diag(off,-1))/dr2
//Make KE Matrix
K_matrix=-hbar^2/(2*me)*SD_matrix

//Make Hamiltonian matrix
H_matrix=K_matrix+U_matrix

//Find Eigenvalues E_n and Eigenfunction psi_N
[e_funct,e_values]=spec(H_matrix)

//All Eigenvalues 1,2,...n where E_N<0
//DISPLAYING ReSULTS(Eigenvalues)
En=diag(e_values)
En=(En(En<0))'
```

```
disp('Quantum state, Energy Eigenvalues(in eV):')
for i=1:length(En)
printf('%i %f\n',i,En(i))
end
nl=pqn-L
printf('Energy Eigenvalue(in eV) for Quntum stae no.%i:(l=%i)\n%f',nl,L,En(nl))

//Normalisation of eigenfunctions
for n=1:nl
   psi(:,n)=[0;e_funct(:,n);0]
   area=inttrap(r,(psi(:,n).*psi(:,n))')
   psi(:,n)=psi(:,n)/sqrt(area)
   prob(:,n)=psi(:,n).*psi(:,n)
end

//PLOTTING RESULTS(Eigenfunctions and probability)

figure(1)
xtitle('probability density vs radius')
xlabel('radius in armstrong')
ylabel('Probabilty Density per meter')
xgrid
plot(r',prob(:,nl),'b')

figure(2)
xtitle('wave function vs radius')
xlabel('radius in armstrong')
ylabel('wave function')
xgrid
plot(r',psi(:,nl),'b')
```

RESULTS AND DISCUSSIONS

After executing the given Scilab code we get the energy eigenvalues and corresponding eigenfunctions the ground state and first excited state

FOR GROUND STATE:

Enter the value of l: 0
Enter the value of n(n>l): 1

Quantum state	Energy Eigenvalues(in eV)
1	-13.612207
2	-3.402503
3	-1.291687

Energy Eigenvalue (in eV) for Quantum state no. 1:(l=0)
-13.612207

GRAPH -1



FOR FIRST EXCITED STATE:

Enter the value of l: 0
Enter the value of n(n>l): 2

Quantum state	Energy Eigenvalues(in eV)
1	-13.612207
2	-3.402503
3	-1.291687

Energy Eigenvalue (in eV) for Quantum state no. 2:(l=0)
-3.402503

GRAPH-1



GRAPH-2



CONCLUSION:

The energy of the ground state and first excited state is -13.612207eV and -3.402503eV respectively.